

In[1]:= Appendix A

In[2]:= Kyle Cranmer, Sven Kreiss, David López - Val, and Tilman Plehn

■ Define log likelihood for Appendix A (Eq. A1 & A2)

```
In[2]:= nu[mu1_, mu2_, s1_, s2_, alpha_, b_] :=
  mu1 * s1 * (1 + eta11 alpha) + mu2 * s2 * (1 + eta12 alpha) + b
In[3]:= nu2[mu1_, mu2_, s1_, s2_, alpha_, b_] :=
  mu1 * s1 * (1 + eta21 alpha) + mu2 * s2 * (1 + eta22 alpha) + b
In[4]:= l[x1_, m1_, s1_, x2_, m2_, s2_, alpha_] :=
  (x1 - m1)^2 / (2 s1^2) + (x2 - m2)^2 / (2 s2^2) + (alpha)^2 / (2 sa^2)
In[5]:= L = l[x1, nu[mu1, mu2, s11, s12, alpha, b1], s1,
  x2, nu2[mu1, mu2, s21, s22, alpha, b2], s2, alpha];
```

■ Solve Conditionl MLE mufix (Equation 2.20)

```
In[6]:= CMLE = Simplify[Solve[D[
  l[x1, nu[mu1, mu2, s11, s12, alpha, b1], s1,
  x2, nu2[mu1, mu2, s21, s22, alpha, b2], s2, alpha], {{mu1, mu2}, 1}] == 0, {mu1, mu2}]]
Out[6]= {{mu1 -> (b2 (s12 + alpha eta12 s12) - b1 (s22 + alpha eta22 s22) +
  s22 x1 + alpha eta22 s22 x1 - s12 x2 - alpha eta12 s12 x2) /
  (- (1 + alpha eta12) (1 + alpha eta21) s12 s21 + (1 + alpha eta11) (1 + alpha eta22) s11 s22),
  mu2 -> (-b2 (s11 + alpha eta11 s11) + b1 (s21 + alpha eta21 s21) -
  s21 x1 - alpha eta21 s21 x1 + s11 x2 + alpha eta11 s11 x2) /
  (- (1 + alpha eta12) (1 + alpha eta21) s12 s21 + (1 + alpha eta11) (1 + alpha eta22) s11 s22)}}
```

■ Solve MLE (Equation A.3)

For some reason, Mathematica has a hard time solving for the MLE, help it out by hand

```
MLE = Simplify[Solve[D[
  l[x1, nu[mu1, mu2, s11, s12, alpha, b1], s1,
  x2, nu2[mu1, mu2, s21, s22, alpha, b2], s2, alpha], {{mu1, mu2}, 1}] == 0, {mu1, mu2, alpha}]]
In[7]:= MLE = {{0, 0, 0}, {CMLE[[1, 1]] /. alpha -> 0, CMLE[[1, 2]] /. alpha -> 0, alpha -> 0}}
Out[7]= {{0, 0, 0},
  {mu1 -> (b2 s12 - b1 s22 + s22 x1 - s12 x2) / (-s12 s21 + s11 s22), mu2 -> (-b2 s11 + b1 s21 - s21 x1 + s11 x2) / (-s12 s21 + s11 s22), alpha -> 0}}
```

■ Equation 2.21 Derivative of CMLE mufix wrt. alpha is - eta_p muhat_p (independent of sa)

```
In[8]:= Eta1PartialSolution = Simplify[D[CMLE[[1, 1, 2]], alpha] / MLE[[2, 1, 2]] /. alpha -> 0]
Out[8]= ((-s12 s21 + s11 s22)
  ((eta12 s12 s21 + eta21 s12 s21 - (eta11 + eta22) s11 s22) (b2 s12 - b1 s22 + s22 x1 - s12 x2) +
  (-s12 s21 + s11 s22) (b2 eta12 s12 - b1 eta22 s22 + eta22 s22 x1 - eta12 s12 x2))) /
  ((s12 s21 - s11 s22)^2 (b2 s12 - b1 s22 + s22 x1 - s12 x2))
In[9]:= Eta2PartialSolution = Simplify[D[CMLE[[1, 2, 2]], alpha] / MLE[[2, 2, 2]] /. alpha -> 0]
Out[9]= ((-s12 s21 + s11 s22)
  ((eta12 s12 s21 + eta21 s12 s21 - (eta11 + eta22) s11 s22) (-b2 s11 + b1 s21 - s21 x1 + s11 x2) +
  (-s12 s21 + s11 s22) (-b2 eta11 s11 + b1 eta21 s21 - eta21 s21 x1 + eta11 s11 x2))) /
  ((s12 s21 - s11 s22)^2 (-b2 s11 + b1 s21 - s21 x1 + s11 x2))
```

Show that Solution in category universal situation is as expected

```
In[10]:= Simplify[Eta1PartialSolution /. {alpha -> 0,
      s11 -> 90, s12 -> 10, x1 -> 100,
      x21 -> 10, s22 -> 90, x2 -> 100,
      eta11 -> eta1, eta21 -> eta1, eta12 -> eta2, eta22 -> eta2}]
```

```
Out[10]:= -eta1
```

- **Equation 2.15 Calculate the fisher info matrices for full likelihood (w/ full dependence or evaluated at MLE)**

```
In[11]:= Ifull = Simplify[D[l[x1, nu[mu1, mu2, s11, s12, alpha, b1], s1,
      x2, nu2[mu1, mu2, s21, s22, alpha, b2], s2, alpha], {{mu1, mu2, alpha}, 2}]];
```

```
In[12]:= IfullAtMLE = Simplify[D[l[x1, nu[mu1, mu2, s11, s12, alpha, b1], s1,
      x2, nu2[mu1, mu2, s21, s22, alpha, b2], s2, alpha], {{mu1, mu2, alpha}, 2}] /. MLE[[2]]
      alpha -> 0];
```

- **Show that the conditional MVG equations are equivalent to $\text{dmu}_p^{\text{fix}} / d\alpha$. Note the independence on sa (uncertainty on α)**

- **Equation 2.23: Calculate the full covariance matrix Sigma**

```
In[13]:= Sigma = Simplify[Inverse[IfullAtMLE]];
```

```
In[14]:= Sigma33 = Sigma[[3, 3]]
```

```
Out[14]:=  $\text{sa}^2$ 
```

Equation 2.22

```
In[15]:= Eta2CovSolution = Simplify[Sigma[[2, 3]] / Sigma[[3, 3]] / MLE[[2, 2, 2]]]
```

```
Out[15]:= (b2 s11 (-eta11 s12 s21 + eta12 s12 s21 + eta21 s12 s21 - eta22 s11 s22) +
      b1 s21 (-eta12 s12 s21 + (eta11 - eta21 + eta22) s11 s22) + eta12 s12 s21^2 x1 -
      eta11 s11 s21 s22 x1 + eta21 s11 s21 s22 x1 - eta22 s11 s21 s22 x1 +
      eta11 s11 s12 s21 x2 - eta12 s11 s12 s21 x2 - eta21 s11 s12 s21 x2 + eta22 s11^2 s22 x2) /
      ((s12 s21 - s11 s22) (-b2 s11 + b1 s21 - s21 x1 + s11 x2))
```

```
In[16]:= Simplify[Eta2PartialSolution - Eta2CovSolution]
```

```
Out[16]:= 0
```

- **Show that the information matrix approach is equivalent**

- Get IfullAtMLE
- get Ieff
- solve for effective etas

Try to show :

- if $I_{\text{main}} = I_{\text{full}}$ for other elements
- if solution to effective etas is the same as the muhat derivative symbolically

Equation 2.15

```
In[17]:= Ieff = Simplify[D[l[x1, nu[mu1, mu2, s11, s12, alpha, b1], s1,
      x2, nu2[mu1, mu2, s21, s22, alpha, b2], s2, alpha], {{mu1, mu2}, 2}]];
```

- **Eq. A6 : Calculate the effective fisher info matrix (w/ full dependence or evaluated at MLE)**

```
In[18]:= IeffAtMLE = Simplify[D[l[x1, nu[mu1, mu2, s11, s12, alpha, b1], s1,
      x2, nu2[mu1, mu2, s21, s22, alpha, b2], s2, alpha], {{mu1, mu2}, 2}] /. MLE[[2]]];
```

In[19]:= **Simplify[Inverse[IeffAtMLE]]**

$$\text{Out[19]} = \left\{ \left\{ \frac{s_{12}^2 s_{22}^2 + s_1^2 s_{22}^2}{(s_{12} s_{21} - s_{11} s_{22})^2}, -\frac{s_{11} s_{12} s_{22}^2 + s_1^2 s_{21} s_{22}}{(s_{12} s_{21} - s_{11} s_{22})^2} \right\}, \right. \\ \left. \left\{ -\frac{s_{11} s_{12} s_{22}^2 + s_1^2 s_{21} s_{22}}{(s_{12} s_{21} - s_{11} s_{22})^2}, \frac{s_{11}^2 s_{22}^2 + s_1^2 s_{21}^2}{(s_{12} s_{21} - s_{11} s_{22})^2} \right\} \right\}$$

■ **Eq 2.9: Define the mu_eff substitution, calculate Jacobian, reparametrize leff -> lmain**

In[20]:= **mueff[mu1_, mu2_, alpha_] := List[mu1 (1 + eta1 alpha), mu2 (1 + eta2 alpha)]**

Eq A8

In[21]:= **J = D[mueff[mu1, mu2, alpha], {{mu1, mu2, alpha}, 1}]**

Out[21]= **{{1 + alpha eta1, 0, eta1 mu1}, {0, 1 + alpha eta2, eta2 mu2}}**

Eq A9

In[22]:= **Imain = Simplify[Transpose[J].IeffAtMLE.J] ;**

In[23]:= **Imain /. alpha -> 0**

$$\text{Out[23]} = \left\{ \left\{ \frac{s_{11}^2}{s_1^2} + \frac{s_{21}^2}{s_2^2}, \frac{s_{11} s_{12}}{s_1^2} + \frac{s_{21} s_{22}}{s_2^2}, \text{eta1 mu1} \left(\frac{s_{11}^2}{s_1^2} + \frac{s_{21}^2}{s_2^2} \right) + \text{eta2 mu2} \left(\frac{s_{11} s_{12}}{s_1^2} + \frac{s_{21} s_{22}}{s_2^2} \right) \right\}, \right. \\ \left\{ \frac{s_{11} s_{12}}{s_1^2} + \frac{s_{21} s_{22}}{s_2^2}, \frac{s_{12}^2}{s_1^2} + \frac{s_{22}^2}{s_2^2}, \text{eta1 mu1} \left(\frac{s_{11} s_{12}}{s_1^2} + \frac{s_{21} s_{22}}{s_2^2} \right) + \text{eta2 mu2} \left(\frac{s_{12}^2}{s_1^2} + \frac{s_{22}^2}{s_2^2} \right) \right\}, \\ \left\{ \text{eta1 mu1} \left(\frac{s_{11}^2}{s_1^2} + \frac{s_{21}^2}{s_2^2} \right) + \text{eta2 mu2} \left(\frac{s_{11} s_{12}}{s_1^2} + \frac{s_{21} s_{22}}{s_2^2} \right), \right. \\ \left. \text{eta1 mu1} \left(\frac{s_{11} s_{12}}{s_1^2} + \frac{s_{21} s_{22}}{s_2^2} \right) + \text{eta2 mu2} \left(\frac{s_{12}^2}{s_1^2} + \frac{s_{22}^2}{s_2^2} \right), \frac{1}{s_1^2 s_2^2} (\text{eta1}^2 \text{mu1}^2 (s_{11}^2 s_{22}^2 + s_1^2 s_{21}^2) + \right. \\ \left. \left. 2 \text{eta1 eta2 mu1 mu2} (s_{11} s_{12} s_{22}^2 + s_1^2 s_{21} s_{22}) + \text{eta2}^2 \text{mu2}^2 (s_{12}^2 s_{22}^2 + s_1^2 s_{22}^2) \right) \right\} \right\}$$

In[24]:= **Iconstr = {{0, 0, 0}, {0, 0, 0}, {0, 0, 1/sa^2}}**

$$\text{Out[24]} = \left\{ \{0, 0, 0\}, \{0, 0, 0\}, \left\{0, 0, \frac{1}{sa^2}\right\} \right\}$$

■ **Show solution to eta's is indeed solution of linear equations RHS=from full model, LHS=reparametrizing L_eff->L_main**

Eq A.4

In[25]:= **RHS13 = IfullAtMLE[[1, 3]] /. alpha -> 0**

$$\text{Out[25]} = \frac{1}{s_1^2 s_2^2 (s_{12} s_{21} - s_{11} s_{22})} \\ (b_1 (-\text{eta12} s_{11} s_{12} s_{22}^2 s_{21} + (\text{eta11} s_{11}^2 s_{22}^2 + (\text{eta21} - \text{eta22}) s_1^2 s_{21}^2) s_{22}) + \\ b_2 (-\text{eta11} s_{11}^2 s_{12} s_{22}^2 + \text{eta12} s_{11}^2 s_{12} s_{22}^2 + s_1^2 s_{21} (-\text{eta21} s_{12} s_{21} + \text{eta22} s_{11} s_{22})) + \\ \text{eta12} s_{11} s_{12} s_{22}^2 s_{21} x_1 - \text{eta11} s_{11}^2 s_{22}^2 s_{22} x_1 - \text{eta21} s_1^2 s_{21}^2 s_{22} x_1 + \text{eta22} s_1^2 s_{21}^2 s_{22} x_1 + \\ \text{eta11} s_{11}^2 s_{12} s_{22}^2 x_2 - \text{eta12} s_{11}^2 s_{12} s_{22}^2 x_2 + \text{eta21} s_1^2 s_{12} s_{21}^2 x_2 - \text{eta22} s_1^2 s_{11} s_{21} s_{22} x_2)$$

```
In[26]:= RHS23 = IfullAtMLE[[2, 3]] /. alpha -> 0
```

$$\text{Out[26]} = \frac{1}{s_1^2 s_2^2 (s_{12} s_{21} - s_{11} s_{22})} \left(b_2 \left(-\eta_{a11} s_{11} s_{12}^2 s_2^2 + \eta_{a12} s_{11} s_{12}^2 s_2^2 + s_1^2 s_{22} (-\eta_{a21} s_{12} s_{21} + \eta_{a22} s_{11} s_{22}) \right) + \right. \\ \left. b_1 \left(-\eta_{a12} s_{12}^2 s_2^2 s_{21} + s_{22} (\eta_{a11} s_{11} s_{12} s_2^2 + (\eta_{a21} - \eta_{a22}) s_1^2 s_{21} s_{22}) \right) + \right. \\ \left. \eta_{a12} s_{12}^2 s_2^2 s_{21} x_1 - \eta_{a11} s_{11} s_{12} s_2^2 s_{22} x_1 - \eta_{a21} s_1^2 s_{21} s_{22}^2 x_1 + \eta_{a22} s_1^2 s_{21} s_{22}^2 x_1 + \right. \\ \left. \eta_{a11} s_{11} s_{12}^2 s_2^2 x_2 - \eta_{a12} s_{11} s_{12}^2 s_2^2 x_2 + \eta_{a21} s_1^2 s_{12} s_{21} s_{22} x_2 - \eta_{a22} s_1^2 s_{11} s_{22}^2 x_2 \right)$$

introduce "unk1" the unknowns we want to solve for

```
In[27]:= LHS13 = Imain[[1, 3]] /. {alpha -> 0, eta1 -> unk1, eta2 -> unk2}
```

$$\text{Out[27]} = \mu_1 \left(\frac{s_{11}^2}{s_1^2} + \frac{s_{21}^2}{s_2^2} \right) \text{unk1} + \mu_2 \left(\frac{s_{11} s_{12}}{s_1^2} + \frac{s_{21} s_{22}}{s_2^2} \right) \text{unk2}$$

```
In[28]:= LHS23 = Imain[[2, 3]] /. {alpha -> 0, eta1 -> unk1, eta2 -> unk2}
```

$$\text{Out[28]} = \mu_1 \left(\frac{s_{11} s_{12}}{s_1^2} + \frac{s_{21} s_{22}}{s_2^2} \right) \text{unk1} + \mu_2 \left(\frac{s_{12}^2}{s_1^2} + \frac{s_{22}^2}{s_2^2} \right) \text{unk2}$$

Eq. A.9 & A.11

```
In[29]:= EtaInfoSolutions = Solve[{LHS13 == RHS13, LHS23 == RHS23}, {unk1, unk2}] /. MLE[[2]] /. {alpha -> 0, unk1 -> eta1, unk2 -> eta2}
```

$$\text{Out[29]} = \left\{ \left\{ \begin{array}{l} \eta_{a1} \rightarrow - \left((-s_{12} s_{21} + s_{11} s_{22}) (b_2 \eta_{a21} s_{12}^2 s_{21} - b_2 \eta_{a11} s_{11} s_{12} s_{22} + b_2 \eta_{a12} s_{11} s_{12} s_{22} - b_2 \eta_{a22} s_{11} s_{12} s_{22} - b_1 \eta_{a12} s_{12} s_{21} s_{22} - b_1 \eta_{a21} s_{12} s_{21} s_{22} + b_1 \eta_{a22} s_{12} s_{21} s_{22} + \right. \right. \\ \left. \left. b_1 \eta_{a11} s_{11} s_{22}^2 + \eta_{a12} s_{12} s_{21} s_{22} x_1 + \eta_{a21} s_{12} s_{21} s_{22} x_1 - \eta_{a22} s_{12} s_{21} s_{22} x_1 - \eta_{a11} s_{11} s_{22}^2 x_1 - \eta_{a21} s_{12}^2 s_{21} x_2 + \right. \right. \\ \left. \left. \eta_{a11} s_{11} s_{12} s_{22} x_2 - \eta_{a12} s_{11} s_{12} s_{22} x_2 + \eta_{a22} s_{11} s_{12} s_{22} x_2) \right) / \right. \\ \left. \left((s_{12} s_{21} - s_{11} s_{22})^2 (b_2 s_{12} - b_1 s_{22} + s_{22} x_1 - s_{12} x_2) \right), \right. \\ \eta_{a2} \rightarrow - \left((-s_{12} s_{21} + s_{11} s_{22}) (b_2 \eta_{a11} s_{11} s_{12} s_{21} - b_2 \eta_{a12} s_{11} s_{12} s_{21} - \right. \\ \left. b_2 \eta_{a21} s_{11} s_{12} s_{21} + b_1 \eta_{a12} s_{12} s_{21}^2 + b_2 \eta_{a22} s_{11}^2 s_{22} - b_1 \eta_{a11} s_{11} s_{21} s_{22} + \right. \\ \left. b_1 \eta_{a21} s_{11} s_{21} s_{22} - b_1 \eta_{a22} s_{11} s_{21} s_{22} - \eta_{a12} s_{12} s_{21}^2 x_1 + \eta_{a11} s_{11} s_{21} s_{22} x_1 - \right. \\ \left. \eta_{a21} s_{11} s_{21} s_{22} x_1 + \eta_{a22} s_{11} s_{21} s_{22} x_1 - \eta_{a11} s_{11} s_{12} s_{21} x_2 + \right. \\ \left. \eta_{a12} s_{11} s_{12} s_{21} x_2 + \eta_{a21} s_{11} s_{12} s_{21} x_2 - \eta_{a22} s_{11}^2 s_{22} x_2) \right) / \\ \left. \left((s_{12} s_{21} - s_{11} s_{22})^2 (-b_2 s_{11} + b_1 s_{21} - s_{21} x_1 + s_{11} x_2) \right) \right\} \end{array} \right\}$$

This shows that eta from derivative is equal to eta from fisher info approach

```
In[30]:= Simplify[Simplify[-D[CMLE[[1, 1, 2]], alpha] / MLE[[2, 1, 2]] /. alpha -> 0] - eta1 /. EtaInfoSolutions[[1]]]
```

```
Out[30]= 0
```

```
In[31]:= Simplify[Simplify[-D[CMLE[[1, 2, 2]], alpha] / MLE[[2, 2, 2]] /. alpha -> 0] - eta2 /. EtaInfoSolutions[[1]]]
```

```
Out[31]= 0
```

Same idea, but in terms of arbitrary elemnts of IeffAtMLE

```
In[32]:= IeffArb = {{Imain11, Imain12}, {Imain12, Imain22}}
```

```
Out[32]= {{Imain11, Imain12}, {Imain12, Imain22}}
```

```

In[33]:= ImainArb = Simplify[Transpose[J].IeffArb.J]

Out[33]= {{(1 + alpha eta1)^2 Imain11, (1 + alpha eta1) (1 + alpha eta2) Imain12,
          (1 + alpha eta1) (eta1 Imain11 mu1 + eta2 Imain12 mu2)},
          {(1 + alpha eta1) (1 + alpha eta2) Imain12, (1 + alpha eta2)^2 Imain22,
          (1 + alpha eta2) (eta1 Imain12 mu1 + eta2 Imain22 mu2)},
          {(1 + alpha eta1) (eta1 Imain11 mu1 + eta2 Imain12 mu2),
          (1 + alpha eta2) (eta1 Imain12 mu1 + eta2 Imain22 mu2),
          eta1^2 Imain11 mu1^2 + 2 eta1 eta2 Imain12 mu1 mu2 + eta2^2 Imain22 mu2^2}}

In[34]:= LHS13Raw = ImainArb[[1, 3]] /. {alpha -> 0, eta1 -> unk1, eta2 -> unk2}

Out[34]= Imain11 mu1 unk1 + Imain12 mu2 unk2

In[35]:= LHS23Raw = ImainArb[[2, 3]] /. {alpha -> 0, eta1 -> unk1, eta2 -> unk2}

Out[35]= Imain12 mu1 unk1 + Imain22 mu2 unk2

In[36]:= Solve[{LHS13Raw == I13, LHS23Raw == I23}, {unk1, unk2}]

Out[36]= {{unk1 -> -\frac{-I23 Imain12 + I13 Imain22}{(Imain12^2 - Imain11 Imain22) mu1}, unk2 -> -\frac{I23 Imain11 - I13 Imain12}{(Imain12^2 - Imain11 Imain22) mu2}}

```

■ **Fig 7: Compare Lfull to recoupled model (from substitution of mueff into the Leff) for Scenarios A,B,C**

```

In[37]:= arg = Simplify[{{MLE[[2, 1, 2]] - mu1 (1 + eta1 alpha), MLE[[2, 2, 2]] - mu2 (1 + eta2 alpha)}} /.
          EtaInfoSolutions[[1]]];

In[38]:= leff = arg.IeffAtMLE.Transpose[arg] / 2 + alpha^2 / (2 sa^2);

Try to profile alpha and repeat

In[39]:= AlphaFullProfile = Solve[D[L, {alpha, 1}] == 0, {alpha}][[1]];

In[40]:= AlphaEffProfile = Solve[D[leff, {alpha, 1}] == 0, {alpha}][[1]];

In[41]:= leffProfiled = leff /. AlphaEffProfile;

In[42]:= LProfiled = L /. AlphaFullProfile;

```

```

In[43]:= ScenarioA = {sa → 1,
  s11 → 45, s12 → 5, b1 → 50, x1 → 100, s1 → 10,
  s21 → 10, s22 → 90, b2 → 0, x2 → 100, s2 → 10,
  eta11 → .2, eta21 → .2, eta12 → .2, eta22 → .2};
ScenarioB = {sa → 1,
  s11 → 45, s12 → 5, b1 → 50, x1 → 100, s1 → 10,
  s21 → 10, s22 → 90, b2 → 0, x2 → 100, s2 → 10,
  eta11 → .1, eta21 → .3, eta12 → .2, eta22 → .2};
ScenarioC = {sa → 1,
  s11 → 45, s12 → 5, b1 → 50, x1 → 100, s1 → 10,
  s21 → 40, s22 → 60, b2 → 0, x2 → 100, s2 → 10,
  eta11 → 0, eta21 → .2, eta12 → 0, eta22 → 0};
ContourPlot[{
  Evaluate[LProfiled /. ScenarioA] == 1.15,
  Evaluate[LProfiled /. ScenarioB] == 1.15,
  Evaluate[LProfiled /. ScenarioC] == 1.15,
  Evaluate[leffProfiled /. ScenarioA] == 1.15,
  Evaluate[leffProfiled /. ScenarioB] == 1.15,
  Evaluate[leffProfiled /. ScenarioC] == 1.15
},
{mu1, .5, 1.8}, {mu2, 0.5, 1.8}, ContourLabels → True,
PlotRange → {{0.5, 1.8}, {0.3, 2.2}}]

```

